Cocol description of Parva (3)

  PRODUCTIONS

    Parva             = "void" identifier "(" ")" Block .

    Block             = "{" { Statement } "}" .

    Statement         =   Block 
                        | ConstDeclarations 
                        | VarDeclarations 
                        | Assignment

                        | IfStatement 
                        | WhileStatement 
                        | ReturnStatement 
                        | HaltStatement

                        | ReadStatement 
                        | WriteStatement

                        | ";" .

    ConstDeclarations = "const" OneConst 
                          { "," OneConst } ";" .

    OneConst          = identifier "=" Constant .

    Constant          =   number | charLit 
                        | "true" | "false" | "null" .

    VarDeclarations   = Type OneVar 
                          { "," OneVar } ";" .

    OneVar            = identifier [ "=" Expression ] .

    Assignment        = Designator "=" Expression ";" .

    Designator        = identifier 
                          [ "[" Expression "]" ] .

    IfStatement       = "if" "(" Condition ")" 
                          Statement .

    WhileStatement    = "while" "(" Condition ")" 
                          Statement .

    ReturnStatement   = "return" ";" .

    HaltStatement     = "halt" ";" .

    ReadStatement     = "read" "(" ReadElement 
                          { "," ReadElement } ")" ";" .

    ReadElement       = stringLit | Designator .

    WriteStatement    = "write" "(" WriteElement 
                          { "," WriteElement } ")" ";" .

      WriteElement      = stringLit | Expression .
